The LinkedResource class is the last, and least used main class. It is mainly
used for creating embedded images. To create an embedded image you will
need to first create a Html formatted AlternateView. Within that alternate view
you create an tag, that points to the ContentId (CID) of the
LinkedResource. You then create a LinkedResource object and add it to the
AlternateView's LinkedResources collection.

Sub EmbedImages()
'create the mail message
Dim mail As New MailMessage()
'set the addresses
mail.From = New MailAddress("me@mycompany.com")
mail.To.Add("you@yourcompany.com")
'set the content
mail.Subject = "This is an email"
'first we create the Plain Text part
Dim plainView As AlternateView = AlternateView.CreateAlternateViewFromString("This is my plain text content, viewable by those clients that don't support html", Nothing, "text/plain")
'then we create the Html part
'to embed images, we need to use the prefix 'cid' in the img src value
'the cid value will map to the Content-Id of a Linked resource.
'thus <img src='cid:companylogo'> will map to a LinkedResource with a ContentId of 'companylogo'
Dim htmlView As AlternateView = AlternateView.CreateAlternateViewFromString("Here is an embedded image.<img src=cid:companylogo>", Nothing, "text/html")
'create the LinkedResource (embedded image)
Dim logo As New LinkedResource("c:\temp\logo.gif")
logo.ContentId = "companylogo"
'add the LinkedResource to the appropriate view
htmlView.LinkedResources.Add(logo)
'add the views
mail.AlternateViews.Add(plainView)
mail.AlternateViews.Add(htmlView)
'send the message
Dim smtp As New SmtpClient("127.0.0.1") 'specify the mail server address
smtp.Send(mail)
End Sub 'EmbedImages